

## Logo: A Computer Language

*Why do we need special languages for computers?*

[Children may have a range of answers. One idea that you want to come into the discussion is that computers work by processing symbols, whereas people learn first through experience and then attach symbols to the experience. Since people make sense of language through experience, they have a capacity to interpret symbols and situations in ways that computers can't.]

[For example, if a waiter asks, "Would you like a glass of orange juice?" and you reply "Yes", you have the expectation that the waiter will go to the kitchen, get you a glass of orange juice, and bring it to you and it is understood by the waiter that you want to drink the orange juice and that you will pay for it. You and the waiter interpret the language used with respect to your experience with waiters, customers, restaurants, and orange juice. Computers don't have this experience. Attempts have been made to give computers background knowledge needed to interpret natural language, but we haven't been able to do so at level that approaches human ability.]

[Another important point for children to understand is that people can make sense of misspellings, double meanings, missing or incorrect punctuation, or unconventional word order, but it is not currently practical to try to get computers to do so when constructing a computer language.]

*OK, so one computer language we have to work with is Logo. Logo was invented back in the 1960s by Seymour Papert and some folks at a company called BBN. I decided to use it with you guys because it was made so that you can easily do interesting things with it right away, but as you learn more and more you can do very complicated things with it. In fact, it is based on a language called LISP, which they use for lots of undergraduate and graduate courses in computer science over at Northwestern.*

[I used Northwestern as an example because it's a well known local university. LISP (or some flavor of it) is used all over the country, so you can use most local or well known universities as examples.]

*Remember how we said that computer languages have to be very specific about the language they use and the way they put words together? Well, the way to think about Logo is that you are text messaging someone who has a very limited vocabulary. This person expects every message you send to be a command, an order to make something happen.*

[Show students how to start up Logo. We had two or three children at each computer.]

*What happens if you give Logo the message,*

? Hello

*Right, Logo sends back a message*

```
I don't know how to Hello
```

*Logo was expecting you to send a command. Logo saw `Hello` and looked it up to see if it was a command he knew how to do. He didn't know how to do it, so it sent back that message.*

*Sometimes Logo understands a word, but doesn't know what to do because he was expecting a command. Try this message:*

```
? 50
You don't say what to do with 50
```

*Logo knows what a 50 is, but doesn't know what to do with 50 because you didn't give Logo a command.*

*One command that Logo understands is `fd`, which is short for "forward". Try typing that in:*

```
? fd
not enough inputs to fd
```

*What's an input? It's information that a command needs in order to do its job. Since Logo understands 50, let's try giving 50 as input:*

```
? fd 50
```

*What happens? We can use 50 as input to other commands. Try these:*

```
? show 50
50
? rt 50
```


*What does `show 50` do? What does `rt 50` do? The word is short for "right". There is also a command called `lt` which is short for "left" and `bk` which is short for "back". What do you think they do? Try them out. See what you can draw.*

[At this point, I let the kids experiment. I walk around the room, giving encouragement and suggestions. I might notice that a particular group could really use `repeat` or would benefit by writing a procedure for something they're doing, so I'll show them how. By the end of day one, every group has done this and the should be shown how to save their work in a file so they can save their procedures.]


## Introduction to Polygons

[Also on day one, we worked through the drawing of squares and rectangles and wrote procedures with variables for drawing squares and rectangles of different sizes. This is important to get them to translate into Logo the idea that similar shapes have the same angles and proportional sides. So for example, I had some of them try to write a procedure for a rectangle that was two times higher than it was wide, but make it so it could be any size, or to draw a letter according to a scale given as input. This will serve them later when drawing right isosceles triangles or 30/60/90 triangles, and gives them another way of looking at similarity and scale.]

[We also did an activity with fraction circles and polygons to help them see how external angles of regular polygons add up to 360 degrees. This helps them with the total trip theorem of turtle geometry. In this activity, children arrange fraction circles around polygons as show below.]



[picture of triangle with fraction circles showing external angles that add up to 360]



[arrangement of fraction circles and polygons from triangle to decagon]

[Younger kids enjoy building the patterns for all the polygons. Middle school kids get the idea after doing one or two and are ready to move on. For a video that describes a version of this activity for younger children, see “A Montessori Approach to Turtle Geometry: The Movie” on the web page

<http://www.leonelearningsystems.com/teachers.htm>]

[We related the fraction circles to turtle turns and measured the fraction circles with a Montessori protractor. So we saw, for instance, that for a pentagon we need to make turns of a little more than 70. With further analysis, we saw that the  $1/5$  fraction circle is  $360 / 5 = 72$  degrees, and wrote a procedure.]

```
to pentagon
repeat 5 [fd 50 rt 360 / 5]
end
```

[If children have not yet been exposed to variables, we can introduce a variable to change the size of the pentagon]

```
to pentagon2 :side
repeat 5 [fd :side rt 360 / 5]
end
```

[Through this and later exercises in handouts, children have the opportunity to discover a general procedure for a polygon]

## Triangle Made by Halving a Square

[This discussion assumes that computers are a limited resource in your classroom. When I did this talk, the student/computer ratio in my classroom was around 2.5 to 1. Much of the time, there were two or three students at the same computer. One way I worked was to take one or two partners from each computer and present one thing, then send them back and grab the other partners to show something else. My goal was to maximize time on computers and also give kids time to struggle with things both individually and collaboratively.]

[This discussion and the following one make a good pair of discussions to split in this way because they are related.]

*Let's start with a square. What do we know about a square?*

[Draw out the fact that all sides are the same length, all angles are 90]




*Suppose we draw one of the diagonals of the square. If we fold this square along the diagonal, do the sides match up?*

[If the child is not sure, she can cut out the square and fold it]

*So if this whole angle is 90 degrees, how much is this angle?*

[Help child identify the 45 degree angles]




*OK. Now let's just look at one of the triangles we created when we drew the diagonal of the square. What do we know about this triangle?*

[Draw out the fact that it is a right isosceles triangle and that the acute angles are 45 degrees. We can confirm the angles are 45 degrees using the fact that the internal angles of a triangle add up to 180 degrees. If necessary, the triangle can be folded to superimpose one 45 degree angle on another to show they are the same.]

*OK, let's erase this half.*

[Erase half of the square so you have a triangle as shown below.]



*So we know these two sides are the same. So if the length of this side is one, what's the length of this side? Right. Now, how can we figure out the length of the third side? What theorem do we know about right triangles?*

[Get to the point of seeing that if the hypotenuse is  $c$  then  $c^2 = 2$ .]

*Hmm... Looks like we need to figure out the square root of 2. Let's try that.*

[If the children are not familiar with the square root of two, try solving for the square root of 2 on the square root board. Note that we reach the limit of the material. Present Papert's demonstration (given in *Mindstorms* in the chapter titled "The Mathematical Unconscious") that the square root of 2 cannot be represented as a fraction. A sample demonstration is given below.]

We want to show that there is no solution for

$$\frac{a}{b} = \sqrt{2}$$

where  $a$  and  $b$  are whole numbers.

[Depending on the age and experience of the children and your learning objectives, you may want them to play with the equation a bit. The point is to get them through a process like the following.]

If we square both sides of our equation, we get

$$\left(\frac{a}{b}\right)^2 = 2$$

$$\left(\frac{a}{b}\right)^2 = \frac{a}{b} \times \frac{a}{b} = \frac{a \times a}{b \times b} = \frac{a^2}{b^2}$$

Since  $\frac{a^2}{b^2} = \left(\frac{a}{b}\right)^2$  and  $\left(\frac{a}{b}\right)^2 = 2$  then

$$\frac{a^2}{b^2} = 2$$

Multiplying both sides of this equation by  $b^2$  we get

$$a^2 = 2b^2$$

Now, we know that we can factor any whole number. What are the prime factors of 30? Right, so we have

$$30 = 2 \times 3 \times 5$$

So the square of 30 is equal to

$$(2 \times 3 \times 5)^2 = (2 \times 3 \times 5) \times (2 \times 3 \times 5) = 2 \times 2 \times 3 \times 3 \times 5 \times 5$$

Do we have an even number of 2's in the factors of 30 squared or an odd number of 2's?

[Do more examples as needed to establish the fact that you will always have an even number of each factor in the square of a number because you double the number of factors when you square]

OK, now let's go back and look at  $a^2 = 2b^2$ . Does  $a^2$  have an even number of 2's or an odd number of 2's in its factors? Right, an even number. What about  $b^2$ ? Right, even. What about  $2b^2$ ?

[You might need to explain this one, even give a concrete example, such as the following]

What if  $b$  is 30? Then

$$2b^2 = 2 \times 30^2 = 2 \times (2 \times 3 \times 5)^2 = 2 \times (2 \times 3 \times 5) \times (2 \times 3 \times 5) = 2 \times 2 \times 2 \times 3 \times 3 \times 5 \times 5$$

[Work through examples as needed for child to understand that  $2b^2$  will always have an odd number of 2's]

If  $a^2$  has an even number of 2's and  $2b^2$  has an odd number of 2's, can  $a^2 = 2b^2$ ?

If it can't be true that  $a^2 = 2b^2$ , can it be true that  $\left(\frac{a}{b}\right)^2 = 2$ ?

So can  $\sqrt{2}$  be a quotient of two whole numbers?

*What does that mean that the square root of 2 cannot be written as a quotient of two whole numbers?*

[Write on the board “The square root of two is not a quotient of two whole numbers”]

*Let’s think of the numbers on a number line. We could draw a line from one side of this page to the other. We could divide that line with tick marks into tenths, hundredths, thousandths, ten thousandths, hundred thousandths, and so on. Suppose each tick mark has a width that is one ten thousandth of the width of the line. Then, if we divided the line into ten thousandths, the whole line would be covered with tick marks.*

*But one of the games that mathematicians like to play is to imagine that things can get smaller and smaller or thinner and thinner forever. Let’s pretend that the tick marks can get as small as we want. So if we’re marking off the line into millionths, the tick marks can be as narrow as a billionth of a line. Or if we mark off the line into billionths, the tick marks can be as narrow as a trillionth of a line. And so on. Forever.*

*If that were the case, then we would never be able to make a tick mark directly over one third. Why? Well, we know that the decimal representation of one third is  $0.\overline{3}$ , with the threes repeating on and on forever. That means that one third is between three tenths and four tenths. It is also between 33 hundredths and 34 hundredths. And it is between 333 thousandths and 334 thousandths, and on and on and on.*

*So, if we could make our tick marks as thin as we wanted, we could make an infinite number of tick marks and there would never be a tick mark directly over one third. Would there be any other empty spaces on our line? What about two thirds? Five sixths? Four sevenths? Looks like there are lots of fractions that would never get a tick mark.*

[In fact, any fraction with a denominator that has a factor other than 2 or 5 cannot be expressed as a terminating decimal, so there are an infinite number of fractions that would never get a tick mark.]

[At this point, refer back to statement written on the board, “The square root of two is not a quotient of two whole numbers”]

*Now, what does it mean that the square root of two is not a quotient of two numbers?*

*It means that if we took all the halves, all the thirds, all the fourths, all the fifths and on and on and on, all the fractions that we could ever write, if our tick marks could get as skinny as we like, we would never draw a tick mark directly over the square root of two. There are other numbers like this. Do you know any?*

[If children haven’t mentioned pi yet (and it was previously introduced), it can be mentioned at this point.]

*All these numbers like the square root of two and pi are called irrational numbers, not because they are crazy numbers, but because they cannot be expressed as the ratio of two whole numbers.*

[There are clearly an infinite number of irrational numbers between 0 and 1 since we have  $\frac{\sqrt{2}}{2}, \frac{\sqrt{2}}{3}, \frac{\sqrt{2}}{4} \dots$ . Need to show that square root of two times a rational number is still irrational]

*There was actually a guy who got into a lot of trouble because of the square root of two. His name was Hippasus. He was a Pythagorean who was born around 500 B.C. Anybody know about the Pythagoreans?*

[Give the children the opportunity to talk about the Pythagoreans, especially the mystical and religious aspects of their beliefs and practices.]

*Well, it turns out that the square root of two was considered a very fundamental and important number by the Pythagoreans. It was against their very deeply held religious beliefs that this number could be irrational. Then Hippasus came along and said, "Hey guys, the square root of two isn't a rational number, and I can prove it." The story goes that this happened when Hippasus and some other Pythagoreans were at sea. After he proved that the square root of 2 was irrational, Pythagoras had him thrown overboard and drowned.*


[You might have a discussion here about others who demonstrated things with reason that challenged the religious beliefs of others (e.g., Galileo and Darwin).]

[The next thing you need to do is talk the students through a construction of a logo procedure to draw the triangle. See the next section for ideas on how that is done.]

[Below is the procedure that draws the triangle]

```
to tri.half.of.square :scale
fd :scale
rt 180 - 45
fd :scale * sqrt 2
rt 180 - 45
fd :scale
rt 90
end
```

[Below is an illustration you might create with the children on the board as you work through the problem.]




[Given that two sides of the right triangle are the same length, we can find the length of the hypotenuse and the turns needed to draw the triangle]

OK, go and show this to your partners, and explain irrational numbers to them.


## Triangle Made by Halving an Equilateral Triangle

*OK, this time, let's start with an equilateral triangle:*



*What do we know about this triangle? What are these angles? If the length of this side is 2, what are the lengths of the other sides?*

*OK, now let's cut this triangle in half by drawing the height:*



*If this angle...*

[point to the top angle before you draw the height]

*...is 60 degrees, what's this angle?*

[point to one of the 30 degree angles created after you draw the height]

*Right, 30 degrees. If the length of each side of the triangle is 2, what is this length here?*

[Indicate length from lower left corner to place where height intersects base]

*Right, 1.*

*OK, let's get rid of this half.*

[erase the right half of the triangle]



*OK, so we know that if this side is 2, then this side is 1. We know that this angle is 30, and this angle is 60. What's this angle [indicate the right angle]? Right, 90. Let's see. Does that make sense? What's 30 plus 60 plus 90? Right 180. Is that what we want?*

*Yes, because these angles should all add up to 180. OK, what's the length of this line? We don't know, but how do we find out? What theorem do we know that tells us about the length of lines in a right triangle? Right, the Pythagorean theorem. What does that theorem say? Right,*

$$a^2 + b^2 = c^2$$

*Which side is our c? Yes, and we said that was how long? 2. Which side will be a? OK, how long did we say that side was? Right, 1. So now we have:*

$$1^2 + b^2 = 2^2$$

*Is that enough information to solve this problem? Let's see. What's 1 squared? What's 2 squared? OK, so we have*

$$1 + b^2 = 4$$

*How do we find out what b is? Good. First we have to subtract 1 from both sides.*

$$b^2 = 3$$

*Now what? Right, take the square of both sides.*

$$b = \sqrt{3}$$

*So b is the square root of 3. Do you think that's a rational number? Why not?*

[Discuss irrational numbers as needed. See previous discussion for more.]

*OK, so we have the lengths and we have these angles.*

[indicate the internal angles]

*Is that all we need to draw this? No, we need to know how much the turtle turns at each corner. We'll figure that out as we write our procedure. What should we call this procedure?*

[Let the children name the procedure. It's good to have discussion about a good name. Make sure they're recording the procedure as you build it, so they can use it later.]

*OK, half.triangle. So what's the first thing we need to write to start off our half.triangle procedure? Right,*

to tri.half.of.equilateral

*Let's say we start down here and go up the height. How far do we go? Right, the square root of 3. So how do we tell the turtle to do that? Right,*

```
fd sqrt 3
```

*Now we're up here at the top. We know this is 30 degrees, so how much does the turtle have to turn? Right. 180 minus 30. Left or right? Yes, left. In fact, we can even write it like this:*

```
lt 180 - 30
```

*and Logo will do the math for us. OK, now we're facing this way. How far do we have to go? Yes, 2. How do we tell the turtle to go forward 2?*

```
fd 2
```

*What's this angle, here? Sixty. So how much do we turn? 180 minus 60. Right or left? Left.*

```
lt 180 - 60
```

*Now what do we need to do? Forward 1.*

```
fd 1
```


*And just to keep things clean, we want the turtle to be facing the same way he was when we started. How much do we need to turn so the turtle is facing straight up again? Right, 90. Left or right? Yup, left.*

```
lt 90
```

*And how do we end the procedure? Right. End.*

```
end
```

*OK, let's see what we have*



```

to tri.half.of.equilateral
fd sqrt 3
lt 180 - 30
fd 2
lt 180 - 60
fd 1
lt 90
end

```

*What do you think? Are we done? Yeah, it looks like this will draw our triangle. What about the size of this thing? Yeah, it's going to draw an awfully small triangle, right? So what can we do? Yes, we can add a variable for the size.*

*Where's the first place we need to add the variable? Right, on the top line. Where else? Do we need to multiply the angles? Why not? OK, how does this look?*

```
to tri.half.of.equilateral :scale
fd :scale * sqrt 3
lt 180 - 30
fd :scale * 2
lt 180 - 60
fd :scale
lt 90
end
```

*OK, go show this one to your partner. See if you can combine your triangles into one figure and then make designs with that figure.*

## Right Triangles and Trig Functions: Arctan

[Here I split the discussion between the use of arctan and two known side lengths and the use of sin and cos and a known acute angle. For the actan discussion, I started with a 3/4/5 triangle, which was familiar to the students, then showed how arctan can be used to find the angles.]

*OK, so far we've figured out how to draw two different right triangles—the right angled isosceles triangles we get from drawing the diagonal of a square, and the triangles we get when we draw a line along the height of an equilateral triangle.*

*But I think you guys know about another right triangle, right? If we have a right triangle, and one of the legs is 3 and another leg is 4, what is the hypotenuse? Right, 5. So we have the lengths of all the sides. Is that all we need to draw the triangle with Logo? No, we need to know the angle of the turns, too.*

*Well, let's think about it. If we make two different triangles with legs 3, and 4, will they have the same angles? Why?*

[If needed, you can repeat the demonstration with geometry sticks from elementary so that children can see that three-sided figures have fixed angles but  $n$ -sided figures with  $n > 3$  do not have fixed angles.]

*So any right triangle with legs 3 and 4 will have the same angles. So we could just measure the angles of any right triangle with legs 3 and 4 and we would know what the angle was. Let's measure these.*

[Measure the angles]

*Looks like this one is around 37 degrees and this one is around 53. Does that make sense? What's 37 plus 53 plus 90? Is that what we should have?*

*So now we can draw our triangle:*

```
to three.four.five :scale
fd 4 * :scale
lt 180 - 37
fd 5 * :scale
lt 180 - 53
fd 3 * :scale
lt 90
end
```

*That works. There are actually a bunch of triples like (3,4,5) that make right triangles. We call them Pythagorean triples. Here are some of them.*

(3, 4, 5)	(20, 21, 29)	(11, 60, 61)	(13, 84, 85)
(5, 12, 13)	(12, 35, 37)	(16, 63, 65)	(36, 77, 85)
(8, 15, 17)	(9, 40, 41)	(33, 56, 65)	(39, 80, 89)
(7, 24, 25)	(28, 45, 53)	(48, 55, 73)	(65, 72, 97)

*So we could do the same thing with these triangles, right? We could make the triangle with sticks or something and then measure the angles of each of them.*


[Depending on time constraints and interest, children may want to make a table for some or all of the sixteen triangles above, or you might stop at this point.]

*People made tables like these for a long time.*

*Nowadays, programs like Logo have a function that figures out the angles in a right triangle whatever the legs are. In math, the function is called the arctangent. In Logo, we write the function like this:*

arctan

*The input to arctan is a quotient—the length of one leg divided by another. Here's how we would show it on our triangle:*



*And if you check with Logo, you will see that the arctan of  $\frac{3}{4}$  is about 37 and the arctan of  $\frac{4}{3}$  is about 53:*

```
show arctan (3 / 4)
36.869897645844
show arctan (4 / 3)
53.130102354156
```

*What do you think the arctangent of  $\frac{6}{8}$  would be? So if one leg was 6 and the other was 8, what would the hypotenuse be? What would the angles be? Does that make sense? Why?*

*So we could write our Logo procedure for 3/4/5 triangles like this:*

```
to three.four.five.a :scale
fd 4 * :scale
lt 180 - arctan (3 / 4)
fd 5 * :scale
lt 180 - arctan (4 / 3)
fd 3 * :scale
lt 90
end
```

*Now that we know how to use arctan to find our angles, we can write a procedure that works with any right triangle if we give the two legs as input:*

```
to right.tri.a :x :y
fd :y
lt 180 - arctan (:x / :y)
fd sqrt sum sq :x sq :y
lt 180 - arctan (:y / :x)
fd :x
lt 90

to sq :x
op :x * :x
end
```

[Children can combine these triangles to form Heronian triangles. For example, you could line up a (5, 12, 13) triangle and (12, 35, 37) triangle along the sides of length 12 to make a larger triangle with sides 13, 37, and 40. This larger triangle is Heronian because its side lengths and area are rational, but it is not a right triangle.]

[Heron (or Hero) of Alexandria is an interesting character for discussion. He was a Hellenistic engineer who lived from about 10 A.D. to 70 A.D., and his best known invention was the first documented steam engine.]

## Right Triangles and Trig Functions: Sine and Cosine

[I give the following presentation to the students who didn't get the arctangent presentation.]

*OK, you guys have done so much great work that we are ready to use some functions from trigonometry that will help us draw lots of different triangles. Does anybody know what trigonometry is?*


[Through discussion you should elicit or introduce the following:

1. The word "trigonometry" comes from the Greek *trigonon* = three angles and *metron* = measure.
2. Trigonometry is a branch of mathematics that deals with triangles.]

*So far, we've just been working with right triangles, and we're going to continue working with right triangles today.*

[Use discussion with geometry sticks if needed to establish that a right triangle is determined if we know one of its acute angles and the length of the hypotenuse. As we did with the arctangent discussion, suggest that we could make a table using an acute angle and a hypotenuse of 1, then introduce sine and cosine functions]

*So we could draw our triangle like this:*



[With help from students, work out the Logo procedure that follows]

```

to right.tri.sc :theta
fd :sin :theta
rt 180 - (90 - :theta)
fd 1
rt 180 - :theta
fd cos :theta
rt 90
end

```

*Theta is a Greek letter. It is common to use Greek letters for angles. Why do you think that is?*

*OK, so we have a triangle, but it's a really small one, isn't it? It's common to use the variable  $r$  for the length of the hypotenuse, so let's add a variable `:r` to our Logo procedure. Where do we need to add it first? Right, to the top line. Which one of these forward instructions tells us the length of the hypotenuse? Right, `fd 1`, so we need to change that to what? Right, `fd :r`. What else do we need to change? Is that all?*

```

to right.tri.sc.r :r :theta
fd :r * sin :theta
rt 180 - (90 - :theta)
fd :r
rt 180 - :theta
fd :r * cos :theta
rt 90
end

```

[If these children did the right isosceles triangle, ask what they think the sine of 45 is. If they did the 30/60/90 triangle, ask what they think the sine of 30 is. Draw pictures as needed.]

*What kind of designs do you think you can make if you can draw any triangle with its hypotenuse length and one acute angle? What kind of spins could you do? What kind of growth could you have? What else could you do?*

*OK, go explain this to your partner and see what you guys can make. Make sure your partner understands how it works. See how you can combine triangles you can make with triangles your partner can make.*

## Further Topics

Here are topics I didn't have time for but would like to explore with the kids:

### *Law of Sines and Law of Cosines*

Another thing to try is to use Logo to draw oblique triangles using the law of sines or the law of cosines. Some code to do that is shown on the next page.

Kids should be able to work out these procedures with enough discussion of the ASA, SAS, and SSS rules of triangles and the laws of sines and cosines and review of the trig functions.

### *Cngons*

In his book, *Visual Modeling with Logo*, Jim Clayson introduces the CNGON, a "centered ngon" procedure that (1) draws polygons around a central point and (2) takes as input the radius of the polygon. Drawing this figure requires trig, and there are lots of cool designs that can be made with it.

## Programming

The ultimate Logo programming text is Brian Harvey's three volume work, *Computer Science Logo Style*. This text is available for free online at the same web site where you can get Berkeley Logo:

<http://www.cs.berkeley.edu/~bh/>

These books were written for high school freshman computer enthusiasts, and cover a lot of material that is traditionally taught at the college level. Still, there's a lot in these books that is accessible to middle school kids.

These procedures draw triangles given angle-side-angle, side-angle-side, or side-side-side.

```
to tri.asa :angle1 :side :angle2
;; see page 121 of Schaum's Outlines: Trigonometry
;; 11.4 Solution of Oblique Triangles
fd :side
rt 180 - :angle1
fd asa.side :angle2 :side :angle1
rt :angle1 + :angle2
fd asa.side :angle1 :side :angle2
rt 180 - :angle2
end

to tri.sas :side1 :angle :side2
fd :side2
rt 180 - :angle
fd :side1
rt 180 - (sas.angle2 :side1 :angle :side2)
fd sas.side3 :side1 :angle :side2
rt :angle + sas.angle2 :side1 :angle :side2
end

to tri.sss :side1 :side2 :side3
fd :side1
rt 180 - sss.angle :side2 :side3 :side1
fd :side3
rt 180 - sss.angle :side1 :side2 :side3
fd :side2
rt 180 - sss.angle :side3 :side1 :side2
end

to asa.side :angle1 :side :angle2
;; remember that  $\sin(180 - :x) = \sin :x$ 
;; so  $\sin(:angle1 + :angle2)$  is the sin of the third angle
op :side * (sin :angle1) / (sin (:angle1 + :angle2))
end

to sas.angle2 :side1 :angle :side2
op arcsin :side2 * (sin :angle) / sas.side3 :side1 :angle :side2
end

to sas.side3 :side1 :angle :side2
op sqrt ((square :side1) + (square :side2) - (2 * :side1 * :side2 * cos :angle))
end

to square :x
op :x * :x
end

to sss.angle :side1 :side2 :side3
op arccos quotient sss.square.diff :side1 :side2 :side3 minus (product 2 :side2 :side3)
end

to sss.square.diff :side1 :side2 :side3
op difference square :side1 sum square :side2 square :side3
end
```